```
In[24]:= (* We verify Stokes Theorem for the upper hemisphere with
           radius R centered at the origin with F(x,y,z)=(xy,x,yz) *)
         (* parametrization of surface *)
         R = 1
         r = {R Sin[ph] Cos[th], R Sin[ph] Sin[th], R Cos[ph]}
         rth = D[r, th]
         rph = D[r, ph]
         n = Simplify[Cross[rph, rth]]

Out[24]= 1

Out[25]= {Cos[th] Sin[ph], Sin[ph] Sin[th], Cos[ph]}

Out[26]= {-Sin[ph] Sin[th], Cos[th] Sin[ph], 0}

Out[27]= {Cos[ph] Cos[th], Cos[ph] Sin[th], -Sin[ph]}

Out[28]= {Cos[th] Sin[ph]^2, Sin[ph]^2 Sin[th], Cos[ph] Sin[ph]}

In[29]:= (* Flux integral *)
         F = {y x, x, y z}
         cF = Curl[F, {x, y, z}]
         cFOr = cF /. {x → r[[1]], y → r[[2]], z → r[[3]]}
         Simplify[Dot[cFOr, n]]
         Integrate[%, {th, 0, 2 Pi}]
         Integrate[%, {ph, 0, Pi/2}]

Out[29]= {x y, x, y z}

Out[30]= {z, 0, 1 - x}

Out[31]= {Cos[ph], 0, 1 - Cos[th] Sin[ph]}

Out[32]= Cos[ph] Sin[ph]

Out[33]= 2 π Cos[ph] Sin[ph]

Out[34]= π

In[35]:= (* Parametrization of boundary *)
         s = {R Cos[t], R Sin[t], 0}
         st = D[s, t]

Out[35]= {Cos[t], Sin[t], 0}

Out[36]= {-Sin[t], Cos[t], 0}
```

```
In[46]:= (* Line integral *)
         FOs = F /. {x → s[[1]], y → s[[2]], z → s[[3]]}
         Dot[FOs, st]
         Integrate[%, {t, 0, 2 Pi}]

Out[46]= {Cos[t] Sin[t], Cos[t], 0}

Out[47]= Cos[t]^2 - Cos[t] Sin[t]^2

Out[48]= π
```